PharmaSUG China 2017 - 49

One Step to Produce Shift Table by Using PROC REPORT

Haiqiang Luo, PPD Inc., Beijing

ABSTRACT

Shift tables play a very important role in clinical trial analysis. A shift table is a table that displays the number of subjects in different range (e.g. low, normal, or high) or interested grade at baseline and then shift or transition at selected time points or time intervals. The purpose of the shift table is to illustrate the progress of changing from baseline and help to make reasonable inference.

The common programming logic is firstly getting the frequency from PROC SUMMARY (or PROC SQL, PROC FREQ), then calculating percentage in data step, transposing the dataset by PROC TRANSPOSE, and last producing the report by PROC REPORT.

This paper simplifies these tedious procedures and use one step PROC REPORT to produce the shift table from analysis dataset.

INTRODUCTION

All safety related data is collected in a clinical trial study for safety analysis. The laboratory data is one of the most important assessments for safety consideration.

There are several summary tables commonly used to present laboratory results. Descriptive statistics, shift tables and CTCAE summaries are some of widely used ones. The shift table is one of the most frequently requested in a clinical study by statisticians or clinicians. This paper will focus on how to create a laboratory shift table in one step PROC REPORT.

SAMPLE DATA USED TO PRODUCE SHIFT TABLE

To help readers understand the whole process and implement the SAS codes in APPENDIX I, a dummy dataset with two visits and one parameter “Hemoglobin (g/L)” is generated from SASHELP.CLASS. The codes and sample data are shown below:

```
*Dummy USUBJID, TRTA, PARAM, AVISIT, BNRIND, ANRIND, ANL01FL;
data sample0;
    set sashelp.class end=eof;
    USUBJID=name;
         if sex="M" then TRTA="Placebo";
    else if sex="F" then TRTA="Drug A";
    PARAM="Hemoglobin (g/L)";
    AVISIT="Week 4";

         if age<11 then BNRIND="Low    ";
    else if age<14 then BNRIND="Normal";
    else BNRIND="High";

         if age<13 then ANRIND="Low    ";
    else if age<17 then ANRIND="Normal";
    else ANRIND="High";

    if not eof then ANL01FL="Y";
    keep USUBJID TRTA PARAM AVISIT BNRIND ANRIND ANL01FL;
;run;

data sample;
    set
        sample0
        sample0(where=(USUBJID>"H") in=in2)
    ;
    if in2 then do;
        AVISIT="Week 8";
    end;
;run;
```

	USUBJID	TRTA	PARAM	AVISIT	BNRIND	ANRIND	ANL01FL
1	Alfred	Placebo	Hemoglobin (g/L)	Week 4	High	Normal	Y
2	Alice	Drug A	Hemoglobin (g/L)	Week 4	Normal	Normal	Y
3	Barbara	Drug A	Hemoglobin (g/L)	Week 4	Normal	Normal	Y
4	Carol	Drug A	Hemoglobin (g/L)	Week 4	High	Normal	Y
5	Henry	Placebo	Hemoglobin (g/L)	Week 4	High	Normal	Y
6	James	Placebo	Hemoglobin (g/L)	Week 4	Normal	Low	Y
7	Jane	Drug A	Hemoglobin (g/L)	Week 4	Normal	Low	Y
8	Janet	Drug A	Hemoglobin (g/L)	Week 4	High	Normal	Y
9	Jeffrey	Placebo	Hemoglobin (g/L)	Week 4	Normal	Normal	Y
10	John	Placebo	Hemoglobin (g/L)	Week 4	Normal	Low	Y
11	Joyce	Drug A	Hemoglobin (g/L)	Week 4	Normal	Low	Y
12	Judy	Drug A	Hemoglobin (g/L)	Week 4	High	Normal	Y
13	Louise	Drug A	Hemoglobin (g/L)	Week 4	Normal	Low	Y
14	Mary	Drug A	Hemoglobin (g/L)	Week 4	High	Normal	Y
15	Philip	Placebo	Hemoglobin (g/L)	Week 4	High	Normal	Y
16	Robert	Placebo	Hemoglobin (g/L)	Week 4	Normal	Low	Y
17	Ronald	Placebo	Hemoglobin (g/L)	Week 4	High	Normal	Y
18	Thomas	Placebo	Hemoglobin (g/L)	Week 4	Normal	Low	Y
19	William	Placebo	Hemoglobin (g/L)	Week 4	High	Normal	
20	Henry	Placebo	Hemoglobin (g/L)	Week 8	High	Normal	Y
21	James	Placebo	Hemoglobin (g/L)	Week 8	Normal	Low	Y
22	Jane	Drug A	Hemoglobin (g/L)	Week 8	Normal	Low	Y
23	Janet	Drug A	Hemoglobin (g/L)	Week 8	High	Normal	Y
24	Jeffrey	Placebo	Hemoglobin (g/L)	Week 8	Normal	Normal	Y
25	John	Placebo	Hemoglobin (g/L)	Week 8	Normal	Low	Y
26	Joyce	Drug A	Hemoglobin (g/L)	Week 8	Normal	Low	Y
27	Judy	Drug A	Hemoglobin (g/L)	Week 8	High	Normal	Y
28	Louise	Drug A	Hemoglobin (g/L)	Week 8	Normal	Low	Y
29	Mary	Drug A	Hemoglobin (g/L)	Week 8	High	Normal	Y
30	Philip	Placebo	Hemoglobin (g/L)	Week 8	High	Normal	Y
31	Robert	Placebo	Hemoglobin (g/L)	Week 8	Normal	Low	Y
32	Ronald	Placebo	Hemoglobin (g/L)	Week 8	High	Normal	Y
33	Thomas	Placebo	Hemoglobin (g/L)	Week 8	Normal	Low	Y
34	William	Placebo	Hemoglobin (g/L)	Week 8	High	Normal	

Note: The sample data is only used to produce the example layout, which may not be reasonable in medical perspective.

SAMPLE REPORT OF SHIFT TABLE

There are several types of shift table presentation formats and layout. This paper will focus on producing the common shift table shown in below, other formats can be easily produced with some modifications in SAS codes.

Parameter: Hemoglobin (g/L)

	Placebo (N=10)				Drug A (N=9)			
	Baseline				Baseline			
Timepoint	Low	Normal	High	Total	Low	Normal	High	Total
Week 4								
Low	0 (0.0)	4 (44.4)	0 (0.0)	4 (44.4)	0 (0.0)	3 (33.3)	0 (0.0)	3 (33.3)
Normal	0 (0.0)	1 (11.1)	4 (44.4)	5 (55.6)	0 (0.0)	2 (22.2)	4 (44.4)	6 (66.7)
High	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)
Total	0 (0.0)	5 (55.6)	4 (44.4)	9 (100.0)	0 (0.0)	5 (55.6)	4 (44.4)	9 (100.0)
Week 8								
Low	0 (0.0)	4 (50.0)	0 (0.0)	4 (50.0)	0 (0.0)	3 (50.0)	0 (0.0)	3 (50.0)
Normal	0 (0.0)	1 (12.5)	3 (37.5)	4 (50.0)	0 (0.0)	0 (0.0)	3 (50.0)	3 (50.0)
High	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)
Total	0 (0.0)	5 (62.5)	3 (37.5)	8 (100.0)	0 (0.0)	3 (50.0)	3 (50.0)	6 (100.0)

GENERATE SHIFT TABLE

The following section will describe how to produce the table from sample data.

SUBSET SOURCE DATASET, DERIVE RECORD'S CONTRIBUTION PERCENTAGE

Generally, the first thing is to filter the source data with pre-specified criteria, such as select specific lab category or parameters, and select the records that meet the analysis requirement with ANL01FL="Y".

PROC SQL will be used in this paper. At the same time, a new variable will be derived to keep each record's contribution percentage. This variable makes it possible to produce shift table by using one step PROC REPORT.

The basic idea behind this new method is to use the distributive law of multiplication:

$$A*(B + C + D) = AB + AC + AD$$

Think of "A" as factor about denominator, (B + C + D) as frequency. i.e.:

$$A = 100/\text{denominator}$$
$$B + C + D = 1+1+1+...$$

There is: $\textbf{Percentage} = \frac{100}{\text{denominator}} * (1 + 1 + 1 + \cdots) = \frac{100}{\text{denominator}} + \frac{100}{\text{denominator}} + \frac{100}{\text{denominator}} + \cdots$

$\frac{100}{\text{denominator}}$ is each record's contribution percentage. So the idea of calculating percentage can be changed to derive the percentage of each record's contribution to total count by treatment, by parameter and by visit, and then summarize them. The codes and dataset are shown below. The summarizing will be introduced in next section.

```
*Subset dataset and get each record's contribution percentage;
proc sql;
    create table sample_step0 as
    select *, 100/count(*) as per
    from sample
    where ANL01FL="Y"
    group by TRTA, PARAM, AVISIT
;quit;
proc sort; by PARAM; run;
```

	USUBJID	TRTA	PARAM	AVISIT	BNRIND	ANRIND	ANL01FL	per
1	Judy	Drug A	Hemoglobin (g/L)	Week 4	High	Normal	Y	11.11111111111...
2	Louise	Drug A	Hemoglobin (g/L)	Week 4	Normal	Low	Y	11.11111111111...
3	Joyce	Drug A	Hemoglobin (g/L)	Week 4	Normal	Low	Y	11.11111111111...
4	Mary	Drug A	Hemoglobin (g/L)	Week 4	High	Normal	Y	11.11111111111...
5	Alice	Drug A	Hemoglobin (g/L)	Week 4	Normal	Normal	Y	11.11111111111...
6	Carol	Drug A	Hemoglobin (g/L)	Week 4	High	Normal	Y	11.11111111111...
7	Jane	Drug A	Hemoglobin (g/L)	Week 4	Normal	Low	Y	11.11111111111...
8	Barbara	Drug A	Hemoglobin (g/L)	Week 4	Normal	Normal	Y	11.11111111111...
9	Janet	Drug A	Hemoglobin (g/L)	Week 4	High	Normal	Y	11.11111111111...
10	Mary	Drug A	Hemoglobin (g/L)	Week 8	High	Normal	Y	16.66666666666...
11	Louise	Drug A	Hemoglobin (g/L)	Week 8	Normal	Low	Y	16.66666666666...
12	Judy	Drug A	Hemoglobin (g/L)	Week 8	High	Normal	Y	16.66666666666...
13	Joyce	Drug A	Hemoglobin (g/L)	Week 8	Normal	Low	Y	16.66666666666...
14	Janet	Drug A	Hemoglobin (g/L)	Week 8	High	Normal	Y	16.66666666666...
15	Jane	Drug A	Hemoglobin (g/L)	Week 8	Normal	Low	Y	16.66666666666...
16	James	Placebo	Hemoglobin (g/L)	Week 4	Normal	Low	Y	11.11111111111...
17	Philip	Placebo	Hemoglobin (g/L)	Week 4	High	Normal	Y	11.11111111111...
18	Henry	Placebo	Hemoglobin (g/L)	Week 4	High	Normal	Y	11.11111111111...
19	John	Placebo	Hemoglobin (g/L)	Week 4	Normal	Low	Y	11.11111111111...
20	Jeffrey	Placebo	Hemoglobin (g/L)	Week 4	Normal	Normal	Y	11.11111111111...
21	Thomas	Placebo	Hemoglobin (g/L)	Week 4	Normal	Low	Y	11.11111111111...
22	Alfred	Placebo	Hemoglobin (g/L)	Week 4	High	Normal	Y	11.11111111111...
23	Ronald	Placebo	Hemoglobin (g/L)	Week 4	High	Normal	Y	11.11111111111...
24	Robert	Placebo	Hemoglobin (g/L)	Week 4	Normal	Low	Y	11.11111111111...
25	Robert	Placebo	Hemoglobin (g/L)	Week 8	Normal	Low	Y	12.5
26	Philip	Placebo	Hemoglobin (g/L)	Week 8	High	Normal	Y	12.5
27	John	Placebo	Hemoglobin (g/L)	Week 8	Normal	Low	Y	12.5
28	Ronald	Placebo	Hemoglobin (g/L)	Week 8	High	Normal	Y	12.5
29	Jeffrey	Placebo	Hemoglobin (g/L)	Week 8	Normal	Normal	Y	12.5
30	James	Placebo	Hemoglobin (g/L)	Week 8	Normal	Low	Y	12.5
31	Henry	Placebo	Hemoglobin (g/L)	Week 8	High	Normal	Y	12.5
32	Thomas	Placebo	Hemoglobin (g/L)	Week 8	Normal	Low	Y	12.5

PRODUCE SHIFT TABLE BY USING ONE STEP PROC REPORT

Then the shift table can be produced by the sample code below.

```
*Produce shift table by using one step PROC REPORT;
ods escapechar = "~";
ods listing close;
ods rtf file="One Step to Produce Shift Table.rtf" style=tlout ;
title1 j=l "Parameter: #byval(PARAM)" ;
proc report data=sample_step0 nowd completerows completecols out=aa split="$" style(report)=[outputwidth=100%];;
by PARAM;
    column PARAM AVISIT ("~\ql Timepoint" ANRIND)(TRTA, ("Baseline~R/RTF'\brdrb\brdrs\li100\ni100'" BNRIND), (N per N_per));
    define PARAM--AVISIT /group noprint;
    define ANRIND/group  mlf preloadfmt format=$shift. order=data style=[width=7.8% pretext="  " asis=on] " ";
    define TRTA  /across preloadfmt format=$trt.   order=data " ";
    define BNRIND/across mlf preloadfmt format=$shift. order=data " ";
    define N     /noprint;
    define per   /noprint;
    define N_per /computed style=[width=7.5% just=r rightmargin=2% asis=on] " ";

    compute before AVISIT;
        line @1 " ";
        line @1 AVISIT $;
    endcomp;
    %macro display_percentage();
        compute N_per/ char length=40;
            %do i=1 %to %eval(2*4);
                _c%eval(3+3*&i.)_=put(_c%eval(1+3*&i.)_,count.)||"("||put(coalesce(_c%eval(2+3*&i.)_,0),per.);
            %end;
        endcomp;
    %mend;
    %display_percentage;
;run;
ods rtf close;
ods listing;
```

1. ACROSS variables + N statics, contribution percentage variable per

PROC REPORT creates a column for each value of the variable specified with ACROSS option, which is similar to transpose the data, the commas in the COLUMN statement signals across will be used. Two variables,TRTA and BNRIND, are specified with ACROSS option..

ACROSS is a summary report and SAS wants to analyze numeric variables. Variable N is the number of observations in the input data set that contribute to the value in a cell of the report. Variable per summarize the contribution percentage in a cell of the report

The variables, (N per N_per) , to the right of the comma will be distributed within the across categories. Because N and per have been assigned NOPRINT option, only the computed variable N_per will be displayed. N_per is derived from N and per in compute block (see the details in **3**).

2. COMPLETEROWS, COMPLETECOLS option + PRELOADFMT + MLF

Option COMPLETEROWS displays all possible combinations of the values of the group variables even if one or more of the combinations do not occur in the input data set. Variable ANRIND is defined as a group variable.

Option COMPLETECOLS creates all possible combinations for the values of the across variables even if one or more of the combinations do not occur within the input data set. COMPLETEROWS and COMPLETECOLS are default options which could be switched off by using NOCOMPLETEROWS and NOCOMPLETECOLS.

The PRELOADFMT option in the DEFINE statement ensures that PROC REPORT uses all user-defined format ranges for the combinations of across or group variables, even when a frequency is zero. Format $trt and $shift are defined for variables TRTA, ANRIND and BNRIND.

```
 %let BigN1=10; %let BigN2=9;
proc format;
     picture count
     low-high="009 "
 ;
     picture per(round)
     low-high="009.9)"
 ;
     value $trt(notsorted)
     "Placebo"="Placebo$(N=&BigN1)"
     "Drug A" ="Drug A$(N=&BigN2)"
 ;
     value $shift(notsorted multilabel)
     "Low"    ="Low"
     "Normal"="Normal"
     "High"   ="High"
     "Low", "Normal", "High"="Total"
 ;run;
```

Note: Total number (the big Ns) in each group is manually assigned here since it is not the key point of this paper. Suggest to derive these values from ADSL.

MLF option enables PROC REPORT to use the format label for a given range or for overlapping ranges to create subgroup combinations that use MULTILABEL formatting. So PROC REPORT generates a “Total” group for BNRIND and ANRIND.

ORDER = DATA make sure PROC REPORT orders values according to their order in the proc format statement which with NOTSORTED option.

3. Derive variable N_per in compute block.

The computed variable N_per will be distributed within the across categories.

The manipulated data will be stored in the dataset specified in OUT option. The output data set contains one variable for each column of the report. PROC REPORT tries to use the name of the report item as the name of the corresponding variable in the output data set. When variable is under or over an across variable, the name of the variable is based on the column number (_C4_, _C5_, and so on). So we could use the codes below to calculate variable N_per in compute block.

```
_c6_ = put(_c4_, count.) || "(" || put(coalesce(_c5_, 0), per.);
_c9_ = put(_c7_, count.) || "(" || put(coalesce(_c8_, 0), per.);
_c12_ = put(_c10_, count.) || "(" || put(coalesce(_c11_, 0), per.);
_c15_ = put(_c13_, count.) || "(" || put(coalesce(_c14_, 0), per.);
_c18_ = put(_c16_, count.) || "(" || put(coalesce(_c17_, 0), per.);
_c21_ = put(_c19_, count.) || "(" || put(coalesce(_c20_, 0), per.);
_c24_ = put(_c22_, count.) || "(" || put(coalesce(_c23_, 0), per.);
_c27_ = put(_c25_, count.) || "(" || put(coalesce(_c26_, 0), per.);
```

Then the shift table is got by using the corresponding columns in OUT dataset as below:

Parameter: Hemoglobin (g/L)

	Placebo (N=10)				Drug A (N=9)			
	Baseline				Baseline			
Timepoint	Low	Normal	High	Total	Low	Normal	High	Total
Week 4								
Low	0 (0.0)	4 (44.4)	0 (0.0)	4 (44.4)	0 (0.0)	3 (33.3)	0 (0.0)	3 (33.3)
Normal	0 (0.0)	1 (11.1)	4 (44.4)	5 (55.6)	0 (0.0)	2 (22.2)	4 (44.4)	6 (66.7)
High	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)
Total	0 (0.0)	5 (55.6)	4 (44.4)	9 (100.0)	0 (0.0)	5 (55.6)	4 (44.4)	9 (100.0)
Week 8								
Low	0 (0.0)	4 (50.0)	0 (0.0)	4 (50.0)	0 (0.0)	3 (50.0)	0 (0.0)	3 (50.0)
Normal	0 (0.0)	1 (12.5)	3 (37.5)	4 (50.0)	0 (0.0)	0 (0.0)	3 (50.0)	3 (50.0)
High	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)	0 (0.0)
Total	0 (0.0)	5 (62.5)	3 (37.5)	8 (100.0)	0 (0.0)	3 (50.0)	3 (50.0)	6 (100.0)

out=aa dataset

	PARAM	PARAM2	AVISIT	ANRIND	_C4_	_C5_	_C6_ (N_per)	_C7_ (N)	_C8_ (per)	_C9_ (N_per)	_C10_ (N)	_C11_ (per)	_C12_ (N_per)	_C13_ (N)	_C14_ (per)	_C15_ (N_per)
1	Hemoglobin (g/L)	Hemoglobin (g/L)	Week 4	Low	0		0 (0.0)	4	44.4444...	4 (44.4)	0		0 (0.0)	4	44.44444...	4 (44.4)
2	Hemoglobin (g/L)	Hemoglobin (g/L)	Week 4	Normal	0		0 (0.0)	1	11.1111...	1 (11.1)	4	44.4444...	4 (44.4)	5	55.55555...	5 (55.6)
3	Hemoglobin (g/L)	Hemoglobin (g/L)	Week 4	High	0		0 (0.0)	0		0 (0.0)	0		0 (0.0)	0		0 (0.0)
4	Hemoglobin (g/L)	Hemoglobin (g/L)	Week 4	Total	0		0 (0.0)	5	55.5555...	5 (55.6)	4	44.4444...	4 (44.4)	9	100	9 (100.0)
5	Hemoglobin (g/L)	Hemoglobin (g/L)	Week 8	Low	0		0 (0.0)	4	50	4 (50.0)	0		0 (0.0)	4	50	4 (50.0)
6	Hemoglobin (g/L)	Hemoglobin (g/L)	Week 8	Normal	0		0 (0.0)	1	12.5	1 (12.5)	3	37.5	3 (37.5)	4	50	4 (50.0)
7	Hemoglobin (g/L)	Hemoglobin (g/L)	Week 8	High	0		0 (0.0)	0		0 (0.0)	0		0 (0.0)	0		0 (0.0)
8	Hemoglobin (g/L)	Hemoglobin (g/L)	Week 8	Total	0		0 (0.0)	5	62.5	5 (62.5)	3	37.5	3 (37.5)	8	100	8 (100.0)

CONCLUSION

Shift tables are very helpful in observing changes from one time point to another time point. The detailed steps of producing the shift table in one REPORT procedure have been described in this paper. The full codes are attached in APPENDIX I.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Haiqiang Luo
Enterprise: PPD Inc.
Address: 25th Floor, Raffles City Beijing Office Tower No.1 Dongzhimen South Street, Dongcheng District
City, State ZIP: Beijing, 100007
Work Phone: +86 10 61846057
Fax: +86 10 61846099
E-mail: Haiqiang.Luo@ppdi.com
Web: www.ppdi.com

APPENDIX I: CODE OF DUMMY DATA AND PRODUCE LAB SHIFT TABLE

Note: The SAS product used in this paper is SAS® BASE version 9.3 running on a WINDOWS environment.

```
%let BigN1=10; %let BigN2=9;

proc format;
       picture count
       low-high="009 "
;
       picture per(round)
       low-high="009.9)"
;
       value $trt(notsorted)
       "Placebo"="Placebo$(N=&BigN1)"
       "Drug A" ="Drug A$(N=&BigN2)"
;
       value $shift(notsorted multilabel)
       "Low"    ="Low"
       "Normal"="Normal"
       "High"   ="High"
       "Low", "Normal", "High"="Total"
;run;

*Dummy USUBJID, TRTA, PARAM, AVISIT, BNRIND, ANRIND, ANL01FL;
data sample0;
       set sashelp.class end=eof;
       USUBJID=name;
              if sex="M" then TRTA="Placebo";
       else if sex="F" then TRTA="Drug A";
       PARAM="Hemoglobin (g/L)";
       AVISIT="Week 4";

              if age<11 then BNRIND="Low    ";
       else if age<14 then BNRIND="Normal";
       else BNRIND="High";

              if age<13 then ANRIND="Low    ";
       else if age<17 then ANRIND="Normal";
       else ANRIND="High";

       if not eof then ANL01FL="Y";
       keep USUBJID TRTA PARAM AVISIT BNRIND ANRIND ANL01FL;
;run;
data sample;
       set
              sample0
              sample0(where=(USUBJID>"H") in=in2)
       ;
       if in2 then do;
              AVISIT="Week 8";
       end;
;run;

*Subset dataset and get each record's contribution percentage;
proc sql;
       create table sample_step0 as
       select *, 100/count(*) as per
       from sample
       where ANL01FL="Y"
       group by TRTA, PARAM, AVISIT
;quit;
proc sort; by PARAM; run;
```

```
*Produce shift table by using one step PROC REPORT;
ods escapechar = "~";
ods listing close;
ods rtf file="One Step to Produce Shift Table.rtf" style=tlout ;
title1 j=l "Parameter: #byval(PARAM)" ;
proc report data=sample_step0 nowd  completerows completecols out=aa split="$"
style(report)=[outputwidth=100%];;
by PARAM;
       column PARAM AVISIT ("~\ql Timepoint" ANRIND) (TRTA,
("Baseline~R/RTF'\brdrb\brdrs\li100\ni100'" BNRIND), (N per N_per));
       define PARAM--AVISIT /group noprint;
       define ANRIND/group  mlf preloadfmt format=$shift. order=data
style=[width=7.8% pretext="  " asis=on] " ";
       define TRTA  /across preloadfmt format=$trt.   order=data " ";
       define BNRIND/across mlf preloadfmt format=$shift. order=data " ";
       define N     /noprint;
       define per   /noprint;
       define N_per /computed style=[width=7.5% just=r rightmargin=2% asis=on] " ";

       compute before AVISIT;
              line @1 " ";
              line @1 AVISIT $;
       endcomp;
       %macro display_percentage();
              compute N_per/ char length=40;
                     %do i=1 %to %eval(2*4);

       _c%eval(3+3*&i.)_=put(_c%eval(1+3*&i.)_,count.)||"("||put(coalesce(_c%eval(2+3
*&i.)_,0),per.);
                     %end;
              endcomp;
       %mend;
       %display_percentage;
;run;
ods rtf close;
ods listing;
```